

LÖSUNG

Nachname: _____

Vorname: _____

Matrikelnummer: _____

Aufgabe	Points	Score
Programmanalyse mit Modulen und I/O	20	
Programmieren mit Listen	28	
Datentypen und Funktionen Höherer Ordnung	22	
Auswertung und Typen	20	
Σ	90	

- Sie haben 90 Minuten Bearbeitungszeit.
- Die Klausur besteht aus 4 Aufgaben, in denen Sie 90 Punkte erreichen können.
- Die erreichbaren Punkte sind am Rand notiert.
- Entfernen Sie nicht die Heftklammer.
- Verwenden Sie keinen roten Stift.
- Antworten können in deutscher oder englischer Sprache verfasst werden.

Aufgabe 1: Programmanalyse mit Modulen und I/O

Betrachten Sie folgendes Programm.

```
1 module Main(main) where
2
3 import qualified System.Random as R -- contains randomIO :: IO Integer
4
5 main = do
6   num <- randomIO
7   putStrLn "Try to guess my number"
8   guessingGame (abs num) 1
9
10 guessingGame :: Integer -> Integer -> IO Integer
11 guessingGame num n = do
12   str <- getLine
13   let x = (read str :: Int)
14   if x == num then finish n
15   else do
16     let reason = if x < num then "small" else "large"
17     let answer = "Your guess was too " ++ reason ++ ". Try again"
18     guessingGame num (n + 1)
19
20 finish n = putStrLn $ "You guessed my number using " ++ show n ++ " tries"
```

Das Programm enthält vier Fehler. Davon führen drei zu einem Compile-Fehler und einer zu unerwünschtem Verhalten bei der Programmausführung.

- Identifizieren Sie die Fehler durch Angabe der Zeilennummern,
- durch eine kurze Erklärung, warum ein Fehler vorliegt, und
- geben Sie an, wie man den Fehler beheben kann.

Beachten Sie, dass alle vier Fehler unabhängig von einander sind, also jeweils separate Korrekturen benötigen.

(a) Fehler #1

(5)

Lösung

Zeile 17: logischer Fehler: die Antwort "your guess..." wird nur in Variable `answer` gespeichert, aber nicht ausgegeben;

Lösung: ändere `let answer =` zu `putStrLn $`

(b) Fehler #2

(5)

Lösung

Zeile 6, Compile-Fehler wegen unqualifizierter Nutzung von `randomIO`.

Lösung: ersetze in Zeile 6 `randomIO` durch `R.randomIO`

(c) Fehler #3

(5)

Lösung

Zeile 10: Compile-Fehler, da der Typ von `guessingGame` nicht stimmt.

Lösung: Ersetze `IO Integer` durch `IO ()` in Zeile 10.

(d) Fehler #4

(5)

Lösung

Zeile 13: da `num :: Integer`, und `num` mit `x` verglichen werden, müsste auch `x` den Typ `Integer` haben, also ist das `read str :: Int` falsch.

Lösung: Ersetze `Int` mit `Integer` in Zeile 13.

Aufgabe 2: Programmieren mit Listen

Ein Wort v ist ein Suffix von Wort w , wenn es ein Wort u gibt, so dass $w = uv$. Zum Beispiel hat “ball” die Suffixe “ball”, “all”, “ll”, “l” und “”.

Suffixe können für beliebige Listen definiert werden, z. B. hat $[1, 2, 7]$ die Suffixe $[1, 2, 7]$, $[2, 7]$, $[7]$ und $[]$.

Für die folgenden Programmieraufgaben dürfen Sie beliebige Prelude Funktionen nutzen, außer in Teil (a).

- (a) Definieren Sie eine Haskell-Funktion `suff1`, die die Liste aller Suffixe einer Eingabe-Liste in absteigender Länge berechnet. (4)

Beispiel: `suff1 "ball"` sollte zu `["ball", "all", "ll", "l", ""]` auswerten und der Ausdruck `suff1 [1, 2, 7]` sollte zu `[[1, 2, 7], [2, 7], [7], []]` auswerten.

Für diesen Teil dürfen Sie keine vordefinierten Funktionen auf Listen verwenden, außer die beiden Listen-Konstrukturen.

- (b) Geben Sie den allgemeinsten Typ ihrer Implementierung von `suff1` aus Teil (a) an. (2)

- (c) Betrachten Sie eine Variante der `foldr`-Funktion: (4)

```
-- zur Erinnerung
foldr :: (a -> b -> b) -> b -> [a] -> b
```

```
-- Variante
fold :: (a -> [a] -> b -> b) -> b -> [a] -> b
fold f e [] = e
fold f e (x : xs) = f x xs (fold f e xs)
```

Definieren Sie eine alternative Funktion `suff2` zur Berechnung der Liste aller Suffixe einer Eingabe-Liste. Im Gegensatz zu `suff1` muss `suff2` auf `fold` basieren und darf keine Rekursion verwenden.

- (d) Im weiteren finden Sie vier Vorschläge `suff3a`–`suff3d` für die Berechnung aller Suffixe einer Liste. Genau eine davon ist richtig, d.h., liefert die Suffixe in absteigender Länge. (13)

1. `suff3a xs = [ take n xs | n <- [ 0 .. length xs] ]`
2. `suff3b xs = [ drop i xs | (i,_) <- zip [0 .. ] xs]`
3. `suff3c xs = map (drop xs) (reverse [0 .. length xs])`
4. `suff3d xs = map (flip drop xs) [0 .. length xs]`

Geben Sie für jede Variante an, ob diese korrekt ist oder nicht. Für die fehlerhaften Varianten geben Sie zudem den Grund an.

Zur Erinnerung:

- `flip :: (a -> b -> c) -> (b -> a -> c)`
- `take, drop :: Int -> [a] -> [a]`

- (e) Wozu wertet der Ausdruck `[ x | x : _ <- suff1 xs]` aus, wenn `xs` eine endliche Liste ist? (5)

Gilt dies auch für unendliche Listen?

Lösung

```
suff1 [] = [[]]  
suff1 (x : xs) = (x : xs) : suff1 xs
```

```
suff1 :: [a] -> [[a]]
```

```
suff2 = fold (\ x xs ys -> (x : xs) : ys) [[]]
```

```
-- suff3a berechnet Praefixe, keine Suffixe  
-- suff3b ist zu kurz, z.B. gilt suff3b [] = []  
-- suff3c compiliert nicht, da "drop" als 1. Argument eine Zahl und keine Liste braucht  
-- suff3d ist korrekt
```

Der Ausdruck `[ x | x : _ <- suff1 xs ]` wertet zu `xs` selber aus, egal ob `xs` endlich oder unendlich ist.

Aufgabe 3: Datentypen und Funktionen Höherer Ordnung

Betrachten Sie folgende Datentypen.

```
data Date = Date Int Int Integer -- Tag, Monat, Jahr
```

```
data Employee = Person
```

```
  String -- Name
```

```
  Double -- Gehalt
```

```
  Date -- Geburtstag
```

- (a) Definieren Sie eine Funktion `increaseSalary :: Integer -> Employee -> Employee`, die einen ganzzahligen Betrag und einen Angestellten als Eingabe erwartet, und das Gehalt dieses Angestellten um den Betrag erhöht. (4)

Lösung

```
increaseSalary x (Person n s d) = Person n (s + fromInteger x) d
```

- (b) Definieren Sie eine Funktion `salaryNov1980 :: [Employee] -> Double`, die die Summe aller Gehälter von Angestellten berechnet, die im November 1980 geboren sind. (5)

Für diesen Teil sind Rekursion und Importe verboten.

Lösung

```
-- Alternative 1
salary1980 xs = sum [ s | Person _ s (Date _ 11 1980) <- xs ]
-- Alternative 2
salary1980 = sum .
  map ( \ (Person _ s _) -> s ) .
  filter ( \ (Person _ _ (Date _ m y)) -> (m,y) == (11,1980))
```

Betrachten Sie nun folgendes Haskell Skript.

```
data Expr = Var String | Num Integer | Sum Expr Expr
```

```
eval :: (String -> Integer) -> Expr -> Integer
```

```
eval a (Var x) = a x
```

```
eval _ (Num n) = n
```

```
eval a (Sum e1 e2) = eval a e1 + eval a e2
```

- (c) Wir möchten einen Ausdruck $e :: \text{Expr}$ vereinfachen. Dabei soll jedes Vorkommen von $\text{Sum } (\text{Num } n1) (\text{Num } n2)$ durch e durch $\text{Num } (n1 + n2)$ ersetzt werden, bis der resultierende Ausdruck das Muster $\text{Sum } (\text{Num } n1) (\text{Num } n2)$ nicht mehr enthält. (6)

Dazu wurde die Funktion $\text{simplify} :: \text{Expr} \rightarrow \text{Expr}$ implementiert.

```
simplify (Var x) = Var x
```

```
simplify (Num n) = Num n
```

```
simplify (Sum (Num n1) (Num n2)) = Num (n1 + n2)
```

```
simplify (Sum e1 e2) = Sum (simplify e1) (simplify e2)
```

Bearbeiten Sie folgende Aufgaben:

- Erfüllt simplify das Ziel? Geben Sie eine ja/nein Antwort.
- Falls die Antwort ja ist, begründen Sie kurz, warum das Resultat von $\text{simplify } e$ kein Vorkommen von $\text{Sum } (\text{Num } n1) (\text{Num } n2)$ enthalten kann.
- Falls die Antwort nein ist, geben Sie einen Beispiel-Ausdruck e vom Typ Expr an, so dass das Ergebnis von $\text{simplify } e$ ein Vorkommen von $\text{Sum } (\text{Num } n1) (\text{Num } n2)$ enthält.

Lösung

Die Antwort ist nein, da die Wiederholung fehlt.

Beispiel:

```
simplify (Sum (Sum (Num 1) (Num 2)) (Num 4))
= Sum (simplify (Sum (Num 1) (Num 2))) (simplify (Num 4))
= Sum (Num 3) (Num 4)
```

- (d) Wir möchten einige Funktionen mit Hilfe einer fold-Funktion für den Datentyp Expr (namens foldExpr) definieren. (7)

```
eval a = foldExpr a id (+)
```

```
vars = foldExpr ( \ x -> [x] ) ( \ _ -> [] ) (++) -- Variablen einer Expr als Liste
```

```
renameVars r = foldExpr (Var . r) Num Sum -- Umbenennung der Variablen
```

Ihre Aufgabe ist es, foldExpr in geeigneter Weise zu definieren, so dass die obigen drei Funktionsdefinitionen korrekt sind.

Lösung

```
foldExpr f g h (Var x) = f x
```

```
foldExpr f g h (Num n) = g n
```

```
foldExpr f g h (Sum e1 e2) = h (foldExpr f g h e1) (foldExpr f g h e2)
```

Aufgabe 4: Auswertung und Typen

Zu jeder Frage gibt es genau eine richtige Antwort. Das Ankreuzen dieser Antwort ergibt 4 Punkte; kein Kreuz zu machen ergibt 1 Punkt; das Ankreuzen einer falschen oder mehrerer Antworten ergibt 0 Punkte.

Setzen Sie Ihre Kreuze wie abgebildet: ☒.

Nutzen Sie ☒, um ein Kreuz zu entfernen, d.h. ☒ ist gleichbedeutend zu ☐.

Betrachten Sie folgendes Programm, was zum Teil aus Definitionen von vordefinierten Funktionen besteht.

```
-- Prelude
take n (x : xs) | n > 0 = x : take (n - 1) xs
take _ _ = []

takeWhile p (x : xs) | p x = x : takeWhile p xs
takeWhile _ _ = []

head (x : _) = x
```

```
-- neue Definition
squares x = x : squares (x * x)
```

- (a) Was ist der allgemeinste Typ von squares? (4)
- ☐ Int -> [Int]
☐ Num a => [a]
☒ Num a => a -> [a]
☐ Num a => Int -> [a]
- (b) Welche Form der Rekursion liegt bei squares vor? (4)
- ☐ linear und end-rekursiv
☒ **linear und guarded**
☐ linear und verschachtelt
☐ nicht-linear
- (c) Was sind die ersten Schritte der Auswertung von take 2 (squares 3) bei lazy Evaluation? (4)
- ☒ = take 2 (3 : squares (3 * 3)) = 3 : take (2 - 1) (squares (3 * 3))
☐ = take 2 (3 : squares (3 * 3)) = take 2 (3 : squares 9)
☐ = take 2 (3 : squares (3 * 3)) = take 2 (3 : 3 * 3 : squares ((3 * 3) * (3 * 3)))
☐ = head (squares 3) : take 1 (tail (squares 3))
- (d) Was passiert, wenn Sie takeWhile (> 0) (squares (3 :: Integer)) in ghci eingeben? (4)
- ☐ Es wird ein Typfehler angezeigt.
☐ Es gibt keine Rückmeldung, und Sie müssen eine unendliche Berechnung mittels CTRL-C abbrechen.
☒ **Sie sehen, wie nach und nach eine Liste mit unendlich vielen Zahlen ausgegeben wird, die Sie mittels CTRL-C abbrechen können.**
☐ Es wird eine endliche Liste erzeugt und angezeigt, da es irgendwann zu einem Overflow bei den Zahlen kommt.
- (e) Was ist das Resultat von head (filter (< 0) (squares (3 :: Int)))? (4)
- ☐ Es wird ein Typfehler angezeigt.
☐ Es gibt keine Rückmeldung, und Sie müssen eine unendliche Berechnung mittels CTRL-C abbrechen.
☐ Es gibt eine Fehlermeldung, weil head [] nicht definiert ist.
☒ **Es wird eine negative Zahl ausgegeben.**